DECUS
PROGRAM LIBRARY

DECUS NO.	8-305
TITLE	PAL III ASSEMBLER OVERLAY FOR CARD READER INPUT
AUTHOR	B. J. Little
COMPANY	Sandia Corporation Livermore, California
DATE	February 25, 1970
SOURCE LANGUAGE	PAL III

PAL III ASSEMBLER OVERLAY FOR CARD READER INPUT

DECUS Program Library Write-up DECUS No. 8-305

TITLE PAL III ASSEMBLER OVERLAY FOR CARD READER INPUT.

ISSUED 2/25/70

AUTHOR B.J. LITTLE

ABSTRACT

THIS OVERLAY REPLACES THE HIGH SPEED READER INPUT WITH CARD READER INPUT. CHARACTER VALIDITY CHECK IS MADE ON EACH CARD COLUMN. DATA LOSS CHECK IS MADE ON EACH CARD. COLUMNS 1-72,PUNCHED WITH IBM 26 PUNCH,ARE PROCESSED BY THE ASSEMBLER. ERROR DIAGNOSTICS ALLOW RECOVERY FROM MOST DATA LOSS,READ ERROR,AND HARDWARE FAILURE SITUATIONS.

REQUIREMENTS

A. COMPUTER--PDP-8/I WITH CR-8/I (GDI MODEL 100 CARD READER).
B. BINARY TAPE--PAL III ASSEMBLER (DEC-08-ASB1-PB DATED 2/24/67).
C. CARD DECK--SOURCE PROGRAM IN PAL III FORMAT.

RESTRICTIONS

ALL 64 CHARACTERS IN HOLLERITH CODE ARE RECOGNIZED AS VALID. ALPHAS, NUMERICS,AND 11 SPECIAL CHARACTERS OF IBM 26 PUNCH CHARACTER SET ARE CONVERTED TO ASCII CODE. REMAINING VALID CODES ARE PROCESSED AS ZEROES (EQUIVALENT TO BLANK TAPE).

USAGE

A. LOAD PAL III OBJECT TAPE WITH BINARY LOADER.
B. LOAD OVERLAY OBJECT TAPE WITH BINARY LOADER.
C. PERFORM ASSEMBLY ON SOURCE PROGRAM USING CARD DECK AND CARD READER INSTEAD OF TAPE AND HIGH SPEED READER.
D. REFER TO ERROR DIAGNOSTICS IN COMMENTS SECTION IF COMPUTER HALTS AT UNEXPECTED POINTS IN THE ASSEMBLY PROCEDURE.

COMMENTS

USER DEFINED SYMBOL TABLE CAPACITY IS 536.

ALL 80 COLUMNS OF THE CARD ARE READ TWICE. FIRST IN THE BINARY MODE THEN THE ALPHA-NUMERIC MODE. THE BINARY READING IS CHECKED FOR CHARACTER VALIDITY. THE ALPHA-NUMERIC READING IS COMPARED TO THE VALID BINARY READING. A CARD COLUMN COUNT IS KEPT AND DATA LOSS IS ASSUMED IF MORE OR LESS THAN 80 COLUMNS ARE READ.

CARRIAGE-RETURN AND LINE-FEED CODES ARE GENERATED AS PROCESSING TERMINATORS FOR EACH CARD. THEY ARE PLACED IMMEDIATELY AFTER THE LAST NON-SPACE CHARACTER PUNCHED IN COLUMNS 1-72 OR IN COLUMNS 1 AND 2 FOR BLANK CARDS.

ERROR DIAGNOSTICS

ACCUMULATOR CONTENTS	PROBABLE CAUSE OF ERROR HALT RECOVERY PROCEDURE
0000	NORMAL HALT IF LAST CARD READ CONTAINED A PAUSE OR $. CARD DONE FLAG--FLAG NOT SET AFTER 80 COLUMNS. RELOAD LAST CARD,PRESS CONT TO RESTART.
0100	DATA LOSS--DATA READY FLAG OCCURED FOR MORE THAN 80 COLUMNS. RELOAD LAST CARD,PRESS CONT TO RESTART.
0200	READ ERROR--INVALID CODE READ IN BINARY MODE OR BINARY AND ALPHA-NUMERIC READINGS DO NOT AGREE. RELOAD LAST CARD,PRESS CONT TO RESTART.
7661-7777 @	DATA LOSS--DATA READY FLAG OCCURED FOR LESS THAN 80 COLUMNS. RELOAD LAST CARD,PRESS CONT TO RESTART.
7660	THERE ARE SEVERAL CAUSES OF AN ERROR HALT WITH THIS ACCUMULATOR INDICATION. ERROR LIGHTS ON CARD READER CONTROL PANEL HELP DETERMINE CAUSE. NO LIGHTS--READ START OR MOTOR START SWITCHES NOT OPERATED. DATA READY FLAG INOPERATIVE IF CARD MOTION WAS OBSERVED. RCSE COMMAND INOPERATIVE IF CARD MOTION WAS ABSENT. ERROR LIGHTS BURNED OUT. HOPPER EMPTY--NORMAL INDICATION WHEN ALL CARDS PLACED IN INPUT HOPPER HAVE BEEN READ. IF PASS IS NOT COMPLETE,LOAD CARDS IN INPUT HOPPER AND PRESS CONT. PICK FAIL--TOO MANY CARDS IN INPUT HOPPER. REMOVE SOME CARDS,PRESS CONT TO RESTART. DARK CHECK--TRANSLUCENT CARD OR READ LAMP INTENSITY TOO HIGH. LAST CARD HAD ALL VALID HOLLERITH CODES. PRESS CONT TO RESTART. DARK CHECK PLUS LIGHT CHECK--TRANSLUCENT CARD OR READ LAMP INTENSITY TOO HIGH. REMOVE CARD FROM READ STATION,RELOAD CARD,PRESS CONT TO RESTART. STACKER FAIL--CARD JAM IN TRACK BETWEEN READ STATION AND STACKER HOPPER. DETERMINE SALVAGE PROCEDURE. SYNC FAIL--THE LAST CARD WAS PROBABLY READ CORRECTLY. TRY TO CONTINUE BY PRESSING CONT. LIGHT CHECK--CARD IN READ STATION TOO LONG. MAY BE ACCOMPANIED BY INCORRECT DATA READINGS DUE TO LOSS OF BITS. IN GENERAL,LOSS OF BITS WILL STILL GENERATE VALID HOLLERITH CODES. PRESSURE ROLLERS IN READ STATION MAY BE WORN OR OUT OF ADJUSTMENT. SAFEST COURSE IS TO RELOAD DECK AND RESTART AT LOCATION 0200.

```
/PROGRAM LISTING
/PAL III ASSEMBLER OVERLAY FOR CARD READER INPUT
/B.J. LITTLE
/  2-25-70

/PAGE ZERO INTERCOMMUNICATION
        *20
AAA,    0               /POINTER TO READ ROUTINE
BBB,    0               /POINTER TO PUNCH ROUTINE
        *25
CHEKI,  CHEK            /CHECKS CHARACTER LEGALITY
        *52
LOPUNI, LOPUN           /ASR 33 PUNCH ROUTINE
HIPUNI, HIPUN           /PP-8/I PUNCH ROUTINE
CDREDI, CDREAD          /CR-8/I CARD READ ROUTINE
NXCOLI, NXTCOL          /NEXT COLUMN
        *111
TEM1,   0               /AVAILABLE TO ALL ROUTINES
TEM2,   0               /...
TEM3,   0               /...
        *124
SF2,    6000            /MASK FOR PASS BITS
RBGN,   0               /READER BUFFER POINTER
RKON,   -120            /80 COLUMNS TO READ
        *131
TBUF,   BFRST           /PASS 3 BUFFER
GOREAD, BFRST           /CARD READ BUFFER
        *136
CHAR,   0               /CURRENT CHARACTER BEING PROCESSED
        *156
SWITCH, 0               /OUT OF CARDS OR READER NOT READY
COLCNT, 0               /NUMBER OF CHARACTERS IN BUFFER
/EQUALITIES
RCSF=6631               /SKIP ON DATA READY
RCRA=6632               /READ ALPHANUMERIC
RCRB=6634               /READ BINARY
RCSD=6671               /SKIP ON CARD DONE FLAG
RCSE=6672               /SELECT CARD READER AND SKIP IF READY
RCRD=6674               /CLEAR CARD DONE FLAG
AUTOA=10
AUTOB=11
AUTOC=12
PASS1=254
PASS2=261
PASS3=266
 IPUN=161
LOPUN=364
CHEK=1506
CR=70
M12=75
M23=76
```

```
/SELECTS CR-8/I CARD READER
/SELECTS PP-8/I HIGH SPEED PUNCH IF ON
/DETERMINES PASS NUMBER FROM SWITCH SETTINGS
        *200
SPAL,   DCA TEM1        /ZERO COUNTER
        PLS             /SELECT PUNCH
        PSF             /PUNCH FLAG SET
        SKP
        JMP .+3         /YES-SELECT READER AND PUNCH
        ISZ TEM1        /WAITED 42 MS
        JMP .-4         /NO
        TAD CDREDI      /USE CARD READER
        DCA AAA         /SET READ ROUTINE POINTER
        TAD GOREAD      /GET CARD BUFFER START
        DCA RBGN        /SET POINTER
        DCA TEM1        /ZERO COUNTER
        ISZ TEM1        /WAIT 18 MS
        JMP .-1
        PSF             /PUNCH FLAG SET
        JMP .+3         /NO
        TAD HIPUNI      /USE HSP (PP-8/I)
        SKP
        TAD LOPUNI      /USE LSP (ASR 33)
        DCA BBB         /SET PUNCH ROUTINE POINTER
PASS,   KCC             /INITIALIZE KEYBOARD
        TLS             /AND TELEPRINTER
        PLS             /AND PUNCH
        CLA OSR
        AND SF2         /PASS NUMBER SET
        SNA
        JMP NXPA        /NO--HALT
        CLL RAL         /GET HIGH ORDER BIT
        SNL             /WAS IT SET
        JMP PASS1       /NO--PASS 1
        SMA CLA         /YES-WHAT ABOUT LOW ORDER
        JMP PASS2       /NOT ON--PASS 2
        JMP PASS3       /ON--PASS 3
NXPA,   HLT
        JMP PASS
        HLT             /FILL 243-250 WITH HALTS
        HLT
        HLT
        HLT
        HLT
        HLT
```

```
/CR-8/I CARD READER ROUTINE
/CALLED BY JMS I AAA
/EXITS WITH LEGAL CHARACTER IN CHAR
        *1400
CDREAD, 0
        ISZ COLCNT   /BUFFER EMPTY
        JMP GETCHR   /NO--GET NEXT CHAR FROM BUFFER
        RCRB         /CLEAR DATA READY FLAG
        CLA CLL
        TAD RBGN     /BEGINNING OF READ BUFFER
        DCA AUTOA
        TAD RKON     /80 COL CONSTANT
        DCA COLCNT
        RCSE         /SKIP IF CARD READER READY
        JMP ENDBF    /GET HELP
        JMP NXTCOL
        *1426
NXTCOL=.
        TAD M23
        DCA TEM1
        DCA TEM2
        RCSF         /SKIP ON DATA READY FLAG
        SKP          /NO DATA--WAIT FOR 2 PICK ATTEMPTS
        JMP I HAVDAT /HAVE DATA
        ISZ TEM2     /INNER DELAY LOOP
        JMP .-4
        ISZ TEM1     /OUTER DELAY LOOP
        JMP .-6
ENDBF=.
        CLA CLL
        TAD COLCNT   /GET ERROR INDICATION
        HLT          /OPERATOR ACTION
        CLA CLL CMA
        DCA COLCNT   /SHOW BUFFER EMPTY
        JMP CDREAD+1
/LOOK FOR LAST NON SPACE CHARACTER,STORE A CR AND LF
LSTCHR, CLA CLL CMA RAL /SET ACC=-2
        TAD AUTOA
        DCA AUTOA    /BACK UP TWO
        TAD I AUTOA /GET CHARACTER
        TAD M240
        SNA CLA      /WAS IT A SPACE
        JMP LSTCHR   /YES
        TAD CR
        DCA I AUTOA /STORE CARRIAGE RETURN
        TAD LF
        DCA I AUTOA /STORE LINE FEED
        TAD RBGN
        CMA IAC      /FORM -RBGN
        TAD AUTOA    /ADD LOC OF BUFFER END
        CMA IAC      /FORM -(NUMBER OF CHARS IN BUFFER)
        DCA COLCNT   /STORE IN COUNTER
        TAD RBGN
        DCA AUTOB    /INITIALIZE BUFFER EMPTYING INDEX
```

```
/PROCESS LAST CARD READ
GETCHR=.
        TAD I AUTOB /GET A CHARACTER
        DCA CHAR
        JMS I CHEKI /IS IT LEGAL
        JMP CDREAD+1 /NO
        JMP I CDREAD /YES
HAVDAT, HVEDTA
M240,   -240
LF,     212
/CARD READ AND PASS 3 BUFFER
        *7150
BFRST=.
        BFRST       /BUFFER START
        *.+122
/CHECK FOR LEGAL CHARACTER,COMPARE ALPHA AND BINARY READ
HVEDTA=.
        RCRB        /READ BINARY
        DCA TEM2    /SAVE CHARACTER
        TAD OCTWD
        DCA AUTOC
        TAD M100
        DCA TEM1
HUNT=.
        TAD I AUTOC /GET OCTAL CODE
        TAD TEM2    /GET CHARACTER
        SNA CLA     /HAVE MATCH
        JMP .+4     /YES
        ISZ TEM1    /THROUGH SEARCHING
        JMP HUNT    /NO
        JMP ERROR   /ILLEGAL CODE
        TAD AUTOC   /GET OCTAL CODE ADDRESS
        TAD M100
        DCA PALFWD  /ALPHA TABLE POINTER
        ISZ TEM1
        JMP .-1
        RCRA        /READ ALPHA-NUMERIC
        TAD ALFWD
        CMA  IAC
        TAD PALFWD
        SZA CLA
        JMP ERROR   /ALPHA AND OCTAL READ DO NOT MATCH
        TAD I PALFWD /GET ASCII CODE
        DCA I AUTOA /STORE IN BUFFER
        ISZ COLCNT  /FINISHED 80 COLS
        JMP I NXCOLI /NO
```

```
/CHECK FOR CARD DONE FLAG AND MORE THAN 80 COLUMNS
        DCA TEM1
        ISZ TEM1     /DELAY FOR CARD DONE FLAG
        SKP
        JMP I ENDBFI /CARD DONE FLAG MISSING
        RCSF
        SKP
        JMP ERROR    /DATA READY FLAG AFTER COL 80
        RCSD         /SKIP ON CARD DONE FLAG
        JMP .-7
        TAD COL73    /SET UP FOR LAST (NON-SPACE) CHARACTER
        DCA AUTOA    /IN COLS 1-72
        JMP I CHRLST /LOOK FOR LAST NON SPACE CHARACTER
/INPUT TERMINATION ON READ ERROR
ERROR=.
        TAD COLCNT
        SNA CLA      /DATA READY AFTER 80 COLUMNS
        JMP OVR80    /YES
        DCA TEM1     /NO--DELAY TO COMPLETE CARD MOTION
        ISZ TEM1
        JMP .-1
        TAD P100
OVR80=.
        TAD P100
        DCA TEM2
        RCRB         /CLEAR DATA READY FLAG
        CLA CLL
        TAD TEM2     /GET ERROR INDICATOR
        HLT          /OPERATOR ACTION
        CLA CLL CMA
        DCA COLCNT   /REMOVE LAST CARD FROM BUFFER
        JMP I .+1
        CDREAD+1     /NEXT CARD
OCTWD,  OCTTAB-1     /OCTAL CODE TABLE START
ALFWD,  ALFTAB       /ASCII CODE TABLE START
PALFWD, 0            /ALPHA TABLE POINTER
COL73,  BFRST+111    /LOCATION IN BUFFER OF COL 73
P100,   100
M100,   -100
MNS40,  -40
ENDBFI, ENDBF        /DUMP THE BUFFER
CHRLST, LSTCHR       /LAST NON-SPACE CHARACTER
```

```
/HOLLERITH TO ASCII CODE TABLE OF 64 DECIMAL CHARACTERS
/LISTED IN 00 TO 77 OCTAL NUMERICAL ORDER CORRESPONDING TO
/INTERNAL CARD READER CODE AS READ IN ALPHA-NUMERIC MODE
/NS(NON-SPECIFIED) CHARACTERS FOR LEGAL HOLLERITH CODES
/PRODUCE 0(EQUIVALENT TO BLANK TAPE) IN ASCII CODE.
        *7400
ALFTAB=.              /CODE CONVERSION FOR ALPHA-NUMERIC MODE
/ASCII CHAR-  HOLLERITH
/CODE  ACTER  CODE
  240  /SPACE BLANK
  261  /1     1
  262  /2     2
  263  /3     3
  264  /4     4
  265  /5     5
  266  /6     6
  267  /7     7
  270  /8     8
  271  /9     9
  0    /NS    8 2
  275  /=     8 3
  247  /'     8 4
  0    /NS    8 5
  0    /NS    8 6
  0    /NS    8 7
  260  /0     0
  257  //     0 1
  323  /S     0 2
  324  /T     0 3
  325  /U     0 4
  326  /V     0 5
  327  /W     0 6
  330  /X     0 7
  331  /Y     0 8
  332  /Z     0 9
  0    /NS    0 8 2
  254  /,     0 8 3
  250  /(     0 8 4
  0    /NS    0 8 5
  0    /NS    0 8 6
  0    /NS    0 8 7
```

/ASCII /CODE	CHAR- ACTER	HOLLERITH CODE
255	/-	11
312	/J	11 1
313	/K	11 2
314	/L	11 3
315	/M	11 4
316	/N	11 5
317	/O	11 6
320	/P	11 7
321	/Q	11 8
322	/R	11 9
0	/NS	11 8 2
244	/$	11 8 3
252	/*	11 8 4
0	/NS	11 8 5
0	/NS	11 8 6
0	/NS	11 8 7
253	/+	12
301	/A	12 1
302	/B	12 2
303	/C	12 3
304	/D	12 4
305	/E	12 5
306	/F	12 6
307	/G	12 7
310	/H	12 8
311	/I	12 9
0	/NS	12 8 2
256	/.	12 8 3
251	/)	12 8 4
0	/NS	12 8 5
0	/NS	12 8 6
0	/NS	12 8 7

```
/HOLLERITH TO OCTAL CODE TABLE OF 64 DECIMAL CHARACTERS
/LISTED IN 00 TO 77 OCTAL NUMERICAL ORDER CORRESPONDING TO
/INTERNAL CARD READER CODE AS READ IN ALPHA-NUMERIC MODE
        *7500
OCTTAB=.            /CODE CONVERSION FOR BINARY MODE
/OCTAL CHAR-  HOLLERITH
/CODE  ACTER  CODE
 0000  /SPACE BLANK
-0400  /1     1
-0200  /2     2
-0100  /3     3
-0040  /4     4
-0020  /5     5
-0010  /6     6
-0004  /7     7
-0002  /8     8
-0001  /9     9
-0202  /NS    8 2
-0102  /=     8 3
-0042  /'     8 4
-0022  /NS    8 5
-0012  /NS    8 6
-0006  /NS    8 7
-1000  /0     0
-1400  //     0 1
-1200  /S     0 2
-1100  /T     0 3
-1040  /U     0 4
-1020  /V     0 5
-1010  /W     0 6
-1004  /X     0 7
-1002  /Y     0 8
-1001  /Z     0 9
-1202  /NS    0 8 2
-1102  /,     0 8 3
-1042  /(     0 8 4
-1022  /NS    0 8 5
-1012  /NS    0 8 6
 1006  /NS    0 8 7
```

```
/OCTAL CHAR-  HOLLERITH
/CODE  ACTER  CODE
-2000  /-     11
-2400  /J     11 1
-2200  /K     11 2
-2100  /L     11 3
-2040  /M     11 4
-2020  /N     11 5
-2010  /O     11 6
-2004  /P     11 7
-2002  /Q     11 8
-2001  /R     11 9
-2202  /NS    11 8 2
-2102  /$     11 8 3
-2042  /*     11 8 4
-2022  /NS    11 8 5
-2012  /NS    11 8 6
 2006  /NS    11 8 7
 4000  /+     12
-4400  /A     12 1
-4200  /B     12 2
 4100  /C     12 3
 4040  /D     12 4
 4020  /E     12 5
-4010  /F     12 6
-4004  /G     12 7
-4002  /H     12 8
-4001  /I     12 9
-4202  /NS    12 8 2
-4102  /.     12 8 3
-4042  /)     12 8 4
-4022  /NS    12 8 5
-4012  /NS    12 8 6
-4006  /NS    12 8 7
       $
```

AAA	ØØ2Ø	CR	ØØ7Ø	M1ØØ	7371	RBGN	Ø125
ALFTAB	74ØØ	ENDBF	144Ø	M12	ØØ75	RCRA	6632
ALFWD	7365	ENDBFI	7373	M23	ØØ76	RCRB	6634
AUTOA	ØØ1Ø	ERROR	7343	M24Ø	1476	RCRD	6674
AUTOB	ØØ11	GETCHR	Ø2ØØ	NXCOLI	ØØ55	RCSD	6671
AUTOC	ØØ12	GOREAD	Ø132	NXPA	Ø241	RCSE	6672
BBB	ØØ21	HAVDAT	1475	NXTCOL	1426	RCSF	6631
BFRST	715Ø	HIPUN	Ø161	OCTTAB	75ØØ	RKON	Ø126
CDREAD	14ØØ	HIPUNI	ØØ53	OCTWD	7364	SF2	Ø124
CDREDI	ØØ54	HUNT	73Ø1	OVR8Ø	7352	SPAL	Ø2ØØ
CHAR	Ø136	HVEDTA	7273	PALFWD	7366	SWITCH	Ø156
CHEK	15Ø6	LF	1477	PASS	Ø224	TBUF	Ø131
CHEKI	ØØ25	LOPUN	Ø364	PASS1	Ø254	TEM1	Ø111
CHRLST	7374	LOPUNI	ØØ52	PASS2	Ø261	TEM2	Ø112
COLCNT	Ø157	LSTCHR	1446	PASS3	Ø266	TEM3	Ø113
COL73	7367	MNS4Ø	7372	P1ØØ	737Ø		

DECUS
PROGRAM LIBRARY

DECUS NO.	8-306
TITLE	LDR - A ONE PASS TRANSPARENT LOADER
AUTHOR	Douglass Henry
COMPANY	Physics-Astronomy Dept. Vanderbuilt University Nashville, Tennessee
DATE	January 20, 1970
SOURCE LANGUAGE	PAL-D

LDR - A ONE PASS TRANSPARENT LOADER

DECUS Program Library Write-up DECUS No. 8-306

ABSTRACT

LDR performs the same function as the loader supplied with the PDP-8 Disc Monitor. It loads 1-4 programs in BIN format from high or low-speed paper tape, DECtape or DECdisk under control of the Disk Monitor. The programs are always loaded in one pass and LDR is completely transparent, meaning the user can load into any part of core (except the monitor area) and that, unless explicitly loaded, every word has the same value after the load process as before.

USAGE

After LDR and .LDR have been LOADed and SAVEd (see below), the user types

. LDR ↵

LDR responds by asking

* IN-

to which the user responds with 1-4 file names in standard Disk Monitor format. (example: S:BIN, D2:BIN2, R:, T:↵) LDR then requests the starting address of the program to be loaded by typing

*ST =

The user answers by typing either ↵ or n ↵, where n is a 1-5 digit octal number. Omission of n is equivalent to n = 07600. If five digits were typed, the first selects the memory field in which execution starts, otherwise field 0 is assumed. Remaining digits give the octal address within a field at which the program is to begin. After the ST = response has been typed, the teletype carriage is returned but is not advanced to a new line. At any point before this, the user may restart a line by typing rubout or may return to the monitor with ↑ C. Later recovery is possible, but harder. (see below)

LDR immediately begins reading, which may force it to run in a wait loop until a device is readied, reads each BIN file in turn, and enters the loaded

program at the start address. If no binary is on paper tape, this should occur without intervention. LDR will halt after each file loaded from paper tape, to give the user time to turn off the reader and, possibly, insert a new tape. To continue, the operator presses CONT and LDR advances to the next file or, if loading is complete, begins execution. In addition, LDR will halt after a file from any device if it detects a check sum error. This can be distinguished from a normal halt because the accumulator will be non-zero, reflecting the amount of the error.

ADDING LDR TO THE DISK MONITOR

Since LDR uses 7 scratch blocks (see below) and only 3 are normally available, SAM_1 must be modified to insure that the extras can be used. To do this, use a program such as DISKLOOK (DECUS 8-111) to examine and modify words 200.167 - 200.172 (i.e., 40370_8 - 40373_8). Each of these words should begin with the octal digits '00' or '01'. If not, the appropriate file(s) must be deleted to obtain this condition. Now modify the four words so that each begins with '01_8' which reserves the area for scratch.

To add LDR as a system program, both .LDR and LDR must be loaded, each in two passes, by the standard loader, then saved. The save commands needed are

```
SAVE .LDR: 7000-7577; 7000
SAVE LDR! 7400; 7525
```

If desired, .LDR can be SAVEd as a system program instead of user, but it is much more difficult to call accidentally if it is a user program.

OPERATION

To be transparent, LDR must save, then restore, the contents of all core locations it uses. Therefore, immediately after the call, it saves

6000_8 - 6777_8, used by .CD., and 7000_8 - 7600_8, used .LDR, on the disk. Since words 7200-7577 were saved in blocks 373_8-374_8 at the last monitor restart, they are not written again, but are moved from one part of the disk to another. The data are not saved in monitor block format, but as an exact core image in disk addresses 76177_8 - 77777_8. LDR then searches DN_1 and SAM_1 for the first block of .LDR, and reads it and the following block into 7000 - 7377. Words 167 - 177 are saved, then .CD., which uses them, is read from disk and entered. After .CD. has read the input devices and returned, 167 - 177 and 6000 - 6777 are restored, the rest of .LDR is read into 7400 - 7577 and .LDR is entered at 7001.

.LDR begins by requesting a starting address, modifying the data returned by .CD. into a more useable form, and changing word 7600 for later use as part of a wait loop. The code that does this may not be re-executed, since it is in the disk/DECtape buffer and may be overwritten. The BIN file(s) are read and loaded in turn directly into core with the exception that instead of placing a data word into field 0 with an address above 6777_8, it is written on disk with an address 70177_8 greater, and an attempt to modify 7600 - 7777 will be ignored. After all binary has been loaded, a read from disk locations 77177 - 77777 into 7000 - 7600 is initiated and a wait loop in locations 7570-7600 is entered. LDR waits until word 7570 is altered from disk, then enters the loaded program. Since the disk operation is continuing when the program is entered, the user must observe certain precautions.

RESTRICTIONS AND HINTS

When the user program is entered, word 7570 has just been read from disk, words 7571 - 7600 remain to be read, and the input operation is still in progress. This means that, assuming 66nsec/word, the user should wait at least 528nsec before halting, using words 7571 - 7600, or doing any disk I/O. In addition,

at the end of this time, there will be an interrupt pending from the disk. The user program is entered with the interrupt off, and, before turning it on, should clear the disk interrupt by a DCMA, DMAR, or DMAW instruction, by using monitor disk I/O routines or by jumping to 7600 to restart the monitor. The easiest, but not the only, way to observe these restrictions is to specify a starting address of 0 or 7600, which results in monitor restart as soon as the load finishes.

The .LDR file must be named in DN_1 and the number of its first block must be below 367_8. This can be checked by examining appropriate parts of the disk, or care can be taken to make .LDR one of the first programs SAVEd. There must be no file whose name is before .LDR in DN_1 and begins with '.L'. The .CD. file must start at its standard location, block 15_8.

In using LDR, unexpected halts and closed loops may result from errors or misuse. There are several routines which read and write the disk, and not all act identically, but, in general, a read error will cause a halt, while a write error loops. A halt will also occur if .LDR is called from the keyboard, and loops will result if the .LDR file is misplaced, if LDR is used to load the binary of .LDR or if the user attempts to restart at 7600_8 after responding to the ST = message. The last is because LDR modifies 7600 and its original contents must be restored before it is executed. The author has found that 7000_8, 7200_8, 7400_8 and 7600_8 all seem to work as well for his installation.

CAUTION

LDR has been written, tested and debugged on a 4K PDP-8/I with one DF32 and one teletype. It should operate as described above on any other configuration supported by the Disk Monitor, but this cannot be guaranteed.